
What is CALL EXECUTE?

• It’s a statement you can call within a data step

• Places code onto the “execution stack” –the 
code is queued to run after the data step has 
completed

• The code can be a string within quotes, or 
contained in a character variable, or an 
expression

• Can “execute” data step code, procedures, 
calls to macros, etc



Is it a macro?

• Although it’s documented within the macro 
facility - no, it’s not really a macro

• But, it can generate SAS code, similar to a 
macro

• And it’s easier to build code
based on the contents of
a dataset

Hmmm, 
smells like 
a macro…

call execute



Conditional %include - macro

File example.sas (in codedir) contains:

This could be conditionally included with a macro, as in:

* Conditionally include some code;

filename codedir "\\dir\Open_Problem_May_2015";

%let x = junk;

%macro include_it(inc_flag);

%if (&inc_flag = 1) %then %do;

%include codedir(example.sas);

%end;

%mend include_it;

%include_it(0);

%put x=&x;

%global x;

%let x = 42;



Conditional %include – call execute

File example.sas (in codedir) again contains:

Can conditionally include with call execute within a data step:

* Conditionally include some code;

filename codedir "\\dir\Open_Problem_May_2015";

%let x = junk;

%let inc_flag = 0;

data _null_;

if (&inc_flag) then call execute('%include codedir(example.sas);');

run;

%put x=&x;

%global x;

%let x = 42;



Create a dataset for each drug
(containing all people with drug for 1+ weeks)

* First, make unique list of drugs;

data all_Rx;

keep Rx;

set Open_Problem_Data_fixed;

do i = 1 to 6;

Rx = scan(treatment, i, '-');

output;

end;

run;

proc sql;

create table uniq_Rx as

select distinct Rx

from all_Rx

order by 1;

quit;

• Creates a dataset with 19 observations

Pre-Step:



Create each of the datasets
using data steps

* Create a dataset for each Rx, containing people with that Rx;

data _null_;

set uniq_Rx;

call execute('data ' || trim(Rx) || ';');

call execute('   set Open_Problem_Data_fixed;');

call execute("   where '-' || treatment contains '-" || trim(Rx) || "';");

call execute('run;');

run;

• Creates 19 datasets, one for each unique prescription in data

• E.g. dataset BISBUTEX has 94 observations, RABUTSIN has 200 observations, etc



16         * Create a dataset for each Rx, containing people with that Rx;

17         

18         data _null_;

19         set uniq_Rx;

20         call execute('data ' || trim(Rx) || ';');

21         call execute('   set Open_Problem_Data_fixed;');

22         call execute("   where '-' || treatment contains '-" || trim(Rx) || "';");

23         call execute('run;');

24         run;

NOTE: There were 19 observations read from the data set WORK.UNIQ_RX.

NOTE: DATA statement used (Total process time):

real time           0.00 seconds

cpu time            0.00 seconds



NOTE: CALL EXECUTE generated line.

1         + data BISBUTEX;

2         +    set Open_Problem_Data_fixed;

3         +    where '-' || treatment contains '-BISBUTEX';

4         + run;

NOTE: There were 94 observations read from the data set WORK.OPEN_PROBLEM_DATA_FIXED.

WHERE '-'||treatment contains '-BISBUTEX';

NOTE: The data set WORK.BISBUTEX has 94 observations and 4 variables.

NOTE: DATA statement used (Total process time):

real time           0.00 seconds

cpu time            0.00 seconds

5         + data BRANDUS;

6         +    set Open_Problem_Data_fixed;

7         +    where '-' || treatment contains '-BRANDUS';

8         + run;

NOTE: There were 55 observations read from the data set WORK.OPEN_PROBLEM_DATA_FIXED.

WHERE '-'||treatment contains '-BRANDUS';

NOTE: The data set WORK.BRANDUS has 55 observations and 4 variables.

NOTE: DATA statement used (Total process time):

real time           0.00 seconds

cpu time            0.01 seconds

9         + data CULLAPON;

10        +    set Open_Problem_Data_fixed;

11        +    where '-' || treatment contains '-CULLAPON';

12        + run;

NOTE: There were 61 observations read from the data set WORK.OPEN_PROBLEM_DATA_FIXED.

WHERE '-'||treatment contains '-CULLAPON';

NOTE: The data set WORK.CULLAPON has 61 observations and 4 variables.

... 



Do the same thing using Proc SQL

* Create a dataset for each Rx, containing people with that Rx;

data _null_;

set uniq_Rx;

call execute('proc sql;');

call execute('create table ' || trim(Rx) || ' as');

call execute('select * from Open_Problem_Data_fixed');

call execute("   where '-' || treatment contains '-" || trim(Rx) || "';");

call execute('quit;');

run;

• Creates 19 datasets, one for each unique prescription in data

• E.g. dataset BRANDUS has 55 observations, ZIBURATOL has 178 observations, etc



NOTE: CALL EXECUTE generated line.

1         + proc sql;

2         + create table BISBUTEX as

3         + select * from Open_Problem_Data_fixed

4         +    where '-' || treatment contains '-BISBUTEX';

NOTE: Table WORK.BISBUTEX created, with 94 rows and 4 columns.

5         + quit;

NOTE: PROCEDURE SQL used (Total process time):

real time           0.02 seconds

cpu time            0.01 seconds

6         + proc sql;

7         + create table BRANDUS as

8         + select * from Open_Problem_Data_fixed

9         +    where '-' || treatment contains '-BRANDUS';

NOTE: Table WORK.BRANDUS created, with 55 rows and 4 columns.

10        + quit;

NOTE: PROCEDURE SQL used (Total process time):

real time           0.00 seconds

cpu time            0.01 seconds

11        + proc sql;

12        + create table CULLAPON as

13        + select * from Open_Problem_Data_fixed

14        +    where '-' || treatment contains '-CULLAPON';

NOTE: Table WORK.CULLAPON created, with 61 rows and 4 columns.

... 



And now with a macro

* Now using a macro (and SQL);

* Create a dataset for each Rx, containing people with that Rx;

%macro create_dataset(drug);

proc sql;

create table &drug as

select * from Open_Problem_Data_fixed

where '-' || treatment contains "-&drug";

quit;

%mend create_dataset;

data _null_;

set uniq_Rx;

call execute('%create_dataset(' || trim(Rx) || ');');

run;

• Creates 19 datasets, one for each unique prescription in data

• E.g. dataset TRYPTIKON has 54 observations, TUTRICCO has 105 observations, etc



Conditionally abort on an error

• In this example, problem duplicate records end up in dataset named problem_dups, if any, 
prior to this step

• Want to terminate the job with an error if any duplicate observations

data _null_;

set problem_dups (obs=1);

* this code is only executed if there were one or more;

* duplicate observations;

call execute('data _null_; abort abend; run;');

run;



If no problems, does nothing

18         

19         data _null_;

20         set problem_dups;

21         * this code is only executed if there were one or more;

22          * duplicate observations;

23         call execute('data _null_; abort abend; run;');

24         run;

NOTE: There were 0 observations read from the data set WORK.PROBLEM_DUPS.

NOTE: DATA statement used (Total process time):

real time           0.00 seconds

cpu time            0.00 seconds



If problems, aborts
Warning: This kills EG connection to server!

NOTE: CALL EXECUTE generated line.

1         + data _null_;  abort abend; run;

ERROR: Execution terminated by an ABORT statement at line 1 column 16, it 

specified the ABEND option.

_ERROR_=1 _N_=1

NOTE: The SAS System stopped processing this step because of errors.

NOTE: DATA statement used (Total process time):

real time           0.00 seconds

cpu time            0.00 seconds



Questions?


